
Subject: Re: 8.* graphics

Posted by [David Fanning](#) on Tue, 17 Jan 2012 18:29:21 GMT

[View Forum Message](#) <> [Reply to Message](#)

David Fanning writes:

```
>
> Russell writes:
>
>> UPDATE: I tried a variant of a trick by David F., but for the y-axis
>> http://www.idlcoyote.com/tips/another\_yaxis.html . For some kooky
>> reason, as soon as I use the axis.pro (function not procedure), the
>> colors and transparency immediately go away. GRRR!!!
>
> Here is some code that worked surprisingly well!
>
> pos = [0.125, 0.15, 0.9, 0.875]
> x = findgen(101)
> y1 = cgDemoData(17)
> y2 = cgDemoData(17)
> cgPlot, x, y1, /NoData, Position=pos, xtickformat='(A1)', $
>   ytickformat='(A1)'
> cgColorfill, [x[0],x,x[-1]], [!Y.Crange[0], y1,!Y.Crange[0]], $
>   /data, color='yellow'
> p1 = cgsnapshot()
> cgerase
> cgPlot, x, y2, /NoData, Position=pos, xtickformat='(A1)', $
>   ytickformat='(A1)'
> cgColorfill, [x[0],x,x[-1]], [!Y.Crange[0], y2,!Y.Crange[0]], $
>   /data, color='olive'
> p2 = cgsnapshot()
> cgblendimage, p1, p2
> cgPlot, x, y1, thick=2, /Noerase, Position=pos
> cgplot, x, y2, /Overplot, thick=2
> END
```

Sorry, I meant to do that with log plots. :-)

```
pos = [0.125, 0.15, 0.9, 0.875]
x = findgen(101)+1
y1 = cgDemoData(17)
y2 = cgDemoData(17)
cgPlot, x, y1, /NoData, Position=pos, xtickformat='(A1)', $
   ytickformat='(A1)', /Xlog, xrange=[1,100]
cgColorfill, [x[0],x,x[-1]], [!Y.Crange[0], y1,!Y.Crange[0]], $
   /data, color='yellow'
p1 = cgsnapshot()
cgerase
```

```
cgPlot, x, y2, /NoData, Position=pos, xtickformat='(A1)', $
    ytickformat='(A1)', /XLog, xrange=[1,100]
cgColorfill, [x[0],x,x[-1]], [!Y.Crange[0], y2,!Y.Crange[0]], $
    /data, color='olive'
p2 = cgsnapshot()
cgblendimage, p1, p2
cgPlot, x, y1, thick=2, /Noerase, Position=pos, /XLog, xrange=[1,100]
cgplot, x, y2, /Overplot, thick=2, /XLog
END
```

Cheers,

David

--

David Fanning, Ph.D.

Fanning Software Consulting, Inc.

Coyote's Guide to IDL Programming: <http://www.idlcoyote.com/>

Sepore ma de ni thui. ("Perhaps thou speakest truth.")
